Vorlesungsergänzung zur Integralrechnung

R. Brigola, WS 2009

Im Folgenden einige Ergänzungen zur Transformation von Integralen für Funktionen einer Variablen.
Diese Ergänzung ist mit Hilfe des Computeralgebra-Systems Maple erstellt.

Veranschaulichung zur Substitutionsregel bei Integralen

```
> restart:with(student):with(plots):
Warning, the name changecoords has been redefined

>
```

Vorgelegt sei das Integral der Funktion f(x)= $x^2 \sqrt{1+x^3}$ **auf dem Intervall [2,3]**

```
> f:=x->x^2*sqrt(1+x^3);
```

$$f := x \to x^2 \sqrt{1+x^3}$$

```
> Int(f(x),x=2..3);
```

$$\int_2^3 x^2 \sqrt{1+x^3}\, dx$$

Mit der Substitution u= $1+x^3$ und $\mathrm{D}(u)(x) = 3\,x^2$ und u(2)=9, u(3)=28 ergibt die Substitutionsregel, dass unser Integral mit dem folgenden Integral übereinstimmt (Prüfen Sie selbst nochmal jeden Schritt nach) . $\mathrm{D}(u)$ steht für die Ableitung von u (English: Derivative)

Mit Maple führt man die Substitution einfach mit dem Befehl *changevar* durch, wobei die Funktion u = u(x) auf dem betrachteten Intervall umkehrbar sein muss.

```
> changevar(u=1+x^3,%,u);
```

$$\int_9^{28} \frac{\sqrt{u}}{3}\, du$$

Nochmalige Substitution mit $\mathrm{u}(v) = v^2$, $\mathrm{D}(u)(v) = 2\,v$, $u^{(-1)}$ (9) = $\sqrt{9}$, $u^{(-1)}$ (28) = $\sqrt{28}$ ergibt

(Wieder selbst nachprüfen)

```
> changevar(v=sqrt(u),%,v);
```

$$\int_{\sqrt{9}}^{\sqrt{28}} \frac{2\,v^2}{3}\,dv$$

Wenn man will, kann man jetzt auch mal den Wert dieses Integrals ausrechnen, hier mit Maple zunächst exakt, danach als dezimale Näherung

```
> value(%);
```

$$\frac{56\sqrt{28}}{9} - 2\sqrt{9}$$

```
> evalf(%);
```

$$26.92490520$$

Nun zur Veranschaulichung dessen, was man da eigentlich tut :

Wenn wir Längeneinheiten im Definitions- und Wertebereich der Funktionen annehmen, so erhalten wir mit den Integralen Flächeninhalte der Gebiete zwischen dem Graph einer integrierten Funktion und Parameterachse.

Das Flächenmaß des Gebiets unterhalb des im Folgenden **blau** gezeichneten Funktionsgraphen der Ausgangsfunktion $x^2\sqrt{1+x^3}$ über dem Intervall [2,3] ergibt dann nach der Substitutionsregel das gleiche Flächenmaß wie die Flächen unterhalb des **rot** gezeichneten transformierten Integranden $\frac{2\,v^2}{3}$, der "niedriger liegt" als der Graph der Ausgangsfunktion, aber ja dafür über ein anderes, hier von Maß her längeres Intervall integriert wird, hier von $\sqrt{9}$ bis $\sqrt{28}$. Gleiches gilt für die Fläche unterhalb des **grün** gezeichneten Graphen der Funktion $\frac{\sqrt{u}}{3}$ über [9,28] (Ergebnis der ersten Substitution oben).

Sie erkennen auch, dass es millionenfach andere Möglichkeiten gibt, den Integranden durch eine andere Funktion zu ersetzen, die über ein passend gewähltes Intervall integriert zu wieder gleichem Flächenmaß führen würde.

Sie erkennen, dass für eine richtige Anwendung der Substitutionsregel
1) der Integrand verändert wird, wodurch in der Regel eine völlig andere Funktion entsteht, und
2) ***zwingend*** *dann das Integrationsintervall so anzupassen ist, dass man nach Integration in beiden Fällen das gleiche Ergebnis erhält.*

Daher kann es bei einer Aufgabe mit Substitution eben keine "halbrichtige" Antwort geben, wenn eine dieser Teilaufgaben nicht richtig gelöst ist, um nochmal die Frage eines Studenten während der heutigen Vorlesung nach der Bewertung von Prüfungsaufgaben aufzugreifen, wenn er etwa zwar die Funktion "richtig" transformiert, aber das Integrationsintervall falsch ist. Ich denke, Sie sehen jetzt am Beispiel sofort, dass man in einem solchen Fall ja ein völlig absurdes "Ergebnis" hätte.

```
> plot1:=plot(x^2*sqrt(1+x^3),x=2..3,color=blue,thickness=2):
> plot2:=plot(sqrt(u)/3,u=9..28,color=green,thickness=2):
> plot3:=plot(2*nu^2/3,nu=sqrt(9)..sqrt(28),color=red,thickness=2):
> plot4:=polygonplot([[3,0],[3,3^2*sqrt(28)]]):
> plot5:=polygonplot([[9,0],[9,1]]):
> plot6:=polygonplot([[28,0], [28,sqrt(28)/3]]):
> plot7:=polygonplot([[sqrt(28),0],[sqrt(28),56/3]]):
> display([plot1,plot2,plot3,plot4,plot5,plot6,plot7]);
```



>

Alle drei Flächeninhalte der Flächenstücke unterhalb des blauen bzw. unterhalb des roten oder grünen Graphen sind gleich groß.

Zum Test vergleichen wir alle drei Integrale: Ihre Differenzen müssen Null ergeben

```
> int(x^2*sqrt(1+x^3),x=2..3)-int(sqrt(u)/3,u=9..28);
```

$$0$$

```
> int(x^2*sqrt(1+x^3),x=2..3)-
int(2*v^2/3,v=sqrt(9)..sqrt(28));
```

$$0$$

Im Folgenden nun zur Demonstration einige Ausführungen zu leider in Übungs- und Klausur-Aufgaben häufig zu sehenden Dilettanten-Fehlern bei der Substitution von Integralen

Wir betrachten zu Demonstrationszwecken folgendes Integral:

```
> restart:with(student):with(plots):
Warning, the name changecoords has been redefined
```

```
> Int(sqrt(x^2+1),x=-1..2);
```

$$\int_{-1}^{2} \sqrt{x^2+1}\, dx$$

Nun könnte man die Idee haben, mit $u = x^2 + 1$ zu substituieren und dann vielleicht rechnen:

$x = \sqrt{u-1}$ und dx = $\frac{1}{2\sqrt{u-1}}$ du und die Integrationsgrenzen mit u(-1)=2 und u(2)=5 angeben.

Dann erhielte man das "Ergebnis":

```
> Int(sqrt(u)/(2*sqrt(u-1)),u=2..5);
```

$$\int_{2}^{5} \frac{\sqrt{u}}{2\sqrt{u-1}}\, du$$

Auch Maple macht das genauso, wenn Sie sagen, sie möchten die Variablen-Substitution u=x^2+1 durchgeführt haben:

```
> changevar(u=x^2+1,Int(sqrt(x^2+1),x=-1..2),u);
```

$$\int_{2}^{5} \frac{\sqrt{u}}{2\sqrt{u-1}}\, du$$

Lassen Sie uns nun prüfen, ob die so durchgeführte Substitution den gleichen Integralwert liefert:

```
> int(sqrt(x^2+1),x=-1..2)-int(sqrt(u)/(2*sqrt(u-
1)),u=2..5);
```

$$-\frac{1}{2}\ln(-2+\sqrt{5})+\sqrt{2}-\frac{1}{2}\ln(\sqrt{2}-1)-\frac{1}{4}\ln(9+4\sqrt{5})+\frac{1}{4}\ln(3+2\sqrt{2})$$

```
> evalf(%);
```

$$2.295587150$$

Die beiden Integrale haben nicht den gleichen Wert! Wo liegt der Fehler?

Nun, der Fehler ist, dass die bei der obigen Substitution verwendete Funktion $u = x^2 + 1$ ***nicht invertierbar auf dem Intervall [-1,2]*** **ist.**
Das war aber Voraussetzung der Substitutionsregel in der verwendeten Art und Weise! Auch die Korrektheit von Maple-Umrechnungen wird nur erreicht, wenn der Benutzer mathematisch korrekte Voraussetzungen selbst gewährleistet !!!

Will man diese Substitution dennoch verwenden, muss man stückweise über Intervallen integrieren, auf denen die Substitution umkehrbar ist. Hier etwa

```
> Int(sqrt(x^2+1),x=-1..2)=Int(sqrt(x^2+1),x=-
1..0)+Int(sqrt(x^2+1),x=0..2);
```

$$\int_{-1}^{2} \sqrt{x^2+1}\, dx = \int_{-1}^{0} \sqrt{x^2+1}\, dx + \int_{0}^{2} \sqrt{x^2+1}\, dx$$

und dann substituieren. Machen wir wieder einen Versuch mit der bei Maple implementierten Variablen-Substitution:

```
> Int(sqrt(x^2+1),x=-
1..0)+Int(sqrt(x^2+1),x=0..2)=changevar(u=x^2+1,Int(sqrt(x^
2+1),x=-
1..0),u)+changevar(u=x^2+1,Int(sqrt(x^2+1),x=0..2),u);
```

$$\int_{-1}^{0} \sqrt{x^2+1}\, dx + \int_{0}^{2} \sqrt{x^2+1}\, dx = \int_{1}^{2} -\frac{\sqrt{u}}{2\sqrt{u-1}}\, du + \int_{1}^{5} \frac{\sqrt{u}}{2\sqrt{u-1}}\, du$$

Leider ist auch das **wieder falsch**, diesmal liegt der *Fehler bei Maple* darin, dass wohl nicht die richtige Umkehrfunktion zu $\mathrm{u}(x) = x^2 + 1$ auf [-1,0] verwendet wird. Diese lautet ja $u^{(-1)}$ (x) = - $\sqrt{x-1}$ (Minuszeichen beachten) und bildet [1,2] bijektiv auf das Intervall [-1,0] ab. Bei Maple (hier Maple 8) wird anscheinend + $\sqrt{x-1}$ verwendet, was bei dem uns vorliegenden linken Zweig der Parabel $x^2 + 1$ nicht die richtige Umkehrfunktion ist.

Das muss man natürlich merken!!! Das erste Integral der linken Seite ergibt sicher eine positive Zahl, das entsprechend erste Integral der rechten Seite dagegen eine negative; daher muss die Transformation falsch sein. Auch Überprüfung mit Maple zeigt sofort, dass die beiden Seiten nicht übereinstimmen:

```
> int(sqrt(x^2+1),x=-1..2);
```

$$\sqrt{5}-\frac{1}{2}\ln(-2+\sqrt{5})+\frac{\sqrt{2}}{2}-\frac{1}{2}\ln(\sqrt{2}-1)$$

```
> int(-sqrt(u)/(2*sqrt(u-1)),u=1..2)+int(sqrt(u)/(2*sqrt(u-1)),u=1..5); #  das ist ja auch wieder unser oben schon falsches Integral von 2 bis 5 nach der fehlerhaften Substitution
```

$$\sqrt{5}+\frac{1}{4}\ln(9+4\sqrt{5})-\frac{\sqrt{2}}{2}-\frac{1}{4}\ln(3+2\sqrt{2})$$

```
> evalf(%%-%); # wieder ist die Differenz nicht Null, das Ergebnis der Substitution also falsch
```

$$2.295587150$$

Hier die Graphik zur Substitution $u=x^2+1$ **auf** [-1,0] (**blau**) **und der zugehörigen Umkehrfunktion** (**rot**)

```
> plot1:=plot(x^2+1,x=-1..0,color=blue, thickness=2): # die Funktion u(x)
> plot2:=plot(-sqrt(u-1),u=1..2,color=red,thickness=2): # die zugehörige Umlehrfunktion
> display({plot1,plot2});
```



Nun machen wir es doch selbst und zwar richtig:

Im Integral

```
> Int(sqrt(x^2+1),x=-1..0);
```

$$\int_{-1}^{0} \sqrt{x^2+1}\, dx$$

substituieren wir $u = x^2 + 1$, d.h. $x = -\sqrt{u-1}$ mit der richtigen Umkehrfunktion (vgl. obige Graphik), d $x = -\frac{1}{2\sqrt{u-1}}$ du , -1= $u^{(-1)}$ (2) und

$0 = u^{(-1)}(1)$ (das Symbol $u^{(-1)}$ bezeichne die Umkehrfunktion zu u). Damit erhalten wir aus der Substitutionsregel nun *das richtige Ergebnis*

```
> Int(sqrt(x^2+1),x=-1..0)=Int(sqrt(u)/(2*sqrt(u-1)),u=1..2);
```

$$\int_{-1}^{0} \sqrt{x^2+1}\, dx = \int_{1}^{2} \frac{\sqrt{u}}{2\sqrt{u-1}}\, du$$

```
> int(sqrt(x^2+1),x=-1..0)-int(sqrt(u)/(2*sqrt(u-1)),u=1..2); #  Die Differenz der beiden Integralwerte sollte Null sein, wenn alles korrekt ist:
```

$$-\frac{1}{2}\ln(\sqrt{2}-1)-\frac{1}{4}\ln(3+2\sqrt{2})$$

Diese Differenz ist tatsächlich Null. Prüfen Sie selbst als schöne Rechenübung, dass $(\sqrt{2}-1)\sqrt{3+2\sqrt{2}}$ =1 ergibt. Das ist nach den Regeln fürs Rechnen mit Logarithmen äquivalent dazu, dass die obige Zahl Null ist.

(Tip: Rechnen Sie nach, dass $a = \frac{1}{\sqrt{2}-1}$ mit $\sqrt{3+2\sqrt{2}}$ übereinstimmt.)

Das zweite Teil-Integral ist unproblematisch und ergibt mit der gleichen Substitution, hier direkt mit Maple transformiert

```
> Int(sqrt(x^2+1),x=0..2)=changevar(u=x^2+1,Int(sqrt(x^2+1),x=0..2),u);
```

$$\int_{0}^{2} \sqrt{x^2+1}\, dx = \int_{1}^{5} \frac{\sqrt{u}}{2\sqrt{u-1}}\, du$$

```
> value(%);
```

$$\sqrt{5}-\frac{1}{2}\ln(-2+\sqrt{5}) = \sqrt{5}+\frac{1}{4}\ln(9+4\sqrt{5})$$

Test von allem zusammen:

```
> Int(sqrt(x^2+1),x=-1..2)-Int(sqrt(u)/(2*sqrt(u-1)),u=1..2)-Int(sqrt(u)/(2*sqrt(u-1)),u=1..5);
```

$$\int_{-1}^{2} \sqrt{x^2+1}\, dx - \int_{1}^{2} \frac{\sqrt{u}}{2\sqrt{u-1}}\, du - \int_{1}^{5} \frac{\sqrt{u}}{2\sqrt{u-1}}\, du$$

```
> value(%);
```

$$-\frac{1}{2}\ln(-2+\sqrt{5})-\frac{1}{2}\ln(\sqrt{2}-1)-\frac{1}{4}\ln(3+2\sqrt{2})-\frac{1}{4}\ln(9+4\sqrt{5})$$

```
> simplify(%);
```

$$0$$

Jetzt war also tatsächlich alles richtig gemacht. Wer noch etwas Rechnen üben will, rechne selbst nach, dass die obige Differenz von Logarithmen wirklich Null ergibt.

Welche Lehre ziehen wir aus dem Ganzen? Man muss immer, auch wenn Computeralgebra eine tolle Hilfe ist, selbst gehörig nachdenken, auf der Hut sein, jeden Schritt, jedes Resultat hinterfragen, die Voraussetzungen beachten, unter denen man "Formeln" anwenden kann, testen, ob's denn stimmen kann. Kurz gesagt: Professionell arbeiten!

Sollten Sie sich schließlich noch für eine Stammfunktion von $\sqrt{x^2+1}$ interessieren: Mit Maple sofort erhältlich durch

```
> int(sqrt(x^2+1),x);
```

$$\frac{x\sqrt{x^2+1}}{2}+\frac{1}{2}\operatorname{arcsinh}(x)$$

```
>
```